

TFE4152 Design of Integrated Circuits

Digital Camera

Sara Roberg Ghabeli, Sigurd Hagen Tullander

Fall 2020

# Contents

<b>1</b>	<b>Introduction</b>	<b>4</b>
<b>2</b>	<b>Theory</b>	<b>4</b>
2.1	Analog circuit . . . . .	4
2.1.1	Conceptual operation of pixel circuit . . . . .	6
2.1.2	Dimensions for switch transistors . . . . .	7
2.1.3	Value for $C_s$ . . . . .	7
2.1.4	Values for M3 and active load . . . . .	8
2.1.5	Process variations . . . . .	8
2.2	Digital camera controller . . . . .	9
2.2.1	Finite state machine . . . . .	9
2.2.2	Output timing . . . . .	11
<b>3</b>	<b>Results</b>	<b>12</b>
3.1	Analog pixel circuit . . . . .	12
3.1.1	Values and dimensions . . . . .	12
3.1.2	Analog circuit simulation . . . . .	13
3.1.3	Process variation simulations . . . . .	13
3.2	Digital circuit . . . . .	13
3.2.1	Digital circuit simulation . . . . .	13
<b>4</b>	<b>Discussion</b>	<b>14</b>
4.1	Analog . . . . .	14
4.1.1	Process variation . . . . .	14
<b>5</b>	<b>Summary</b>	<b>15</b>
	<b>Appendices</b>	<b>16</b>

A Analog simulation graphs . . . . .	16
B Spice code . . . . .	20
C Verilog code . . . . .	27

# 1 Introduction

In this report we will address the design, simulation and verification of a 2 x 2 pixels digital camera. That includes mainly 2 systems: A camera controller and a pixel exposure circuit. The camera controller is a fully digital system that take input from 2 buttons to increase and decrease the exposure time, 1 button to reset the camera and 1 button to take a picture. The camera controller will based on this information provide the necessary control signals for the analog circuit to do what it should. The analog circuit is as the name suggests a fully analog system that captures light intensity in capacitors and send data about the image taken to an ADC. What happens with the signal after being sent to the ADC is beyond the scope of this report.

## 2 Theory

### 2.1 Analog circuit

The analog circuit has input signals delivered by the digital circuitry, and delivers its output to two ADCs. The analog circuit diagram is given in figure 1, where each of the four pixel block has the circuit given in figure 2.




Figure 1: Analog circuit schematic.




Figure 2: Pixel circuit schematic.

In detail, the analog circuit consists of 4 pixels in a 2x2 array. Each pixel in the same row shares the same *NRE* (Not-REad) input which means they are sending data to the ADC at the same time. Each pixel in the same column share the same active load and readout wire, which means they need to send data to the ADC at different times.

### 2.1.1 Conceptual operation of pixel circuit

Schematics for the pixel circuit is given in figure 2. It consists of a photo diode PD1, three switch transistors M1, M2 and M4, one buffer transistor M3 and a charge storage capacitor  $C_S$ . The photo diode is modeled by a diode and an ideal current source in parallel. The current source outputs a current  $I_D$  that is proportional to the illumination intensity on that particular pixel. This current is used to charge the transistor  $C_S$ . Ideally, we want that all of  $I_D$  flows through M1 and into  $C_S$  when *EXPOSE* is HIGH and all of  $I_D$  flows back through the diode in the photo diode model when *EXPOSE* is LOW. In that way the voltage on node N2 when *EXPOSE* has been HIGH for a time  $t$  is given by

$$V_{CS}(t) = \frac{1}{C_S} \int_0^t I_D(t) dt. \quad (1)$$

This means the voltage  $V_{CS}$  is proportional to the total illumination on the pixel throughout the exposure time, which is exactly what we want.

But since the camera is supposed to be reusable, we need a way to erase that charge before taking a new picture. That is what M2 is for. Ideally we want M2 to be a perfect switch, so that when *ERASE* is HIGH,  $C_S$  is instantly uncharged through M2 and when *ERASE* is LOW, M2 does not conduct any current at all.

For the readout we do not want to connect N2 directly to OUT, since the output wire might

be very long and have a big capacitance. Instead, the voltage  $V_{CS}$  controls the conductance through a transistor M3. The higher  $V_{CS}$ , the lower conductance through M3 and higher voltage on OUT. In that way, the current required to control the voltage on the output wire is supplied through the active load in the top of each output wire, outside the pixel circuit.

But since all pixels in the same column share the same output wire we need a way to unconnect N3 from OUT, and that is what M4 is for. M4 is supposed to be an ideal switch so that the pixel is driving OUT only when *NRE* is LOW.

### 2.1.2 Dimensions for switch transistors

The transistors M1, M2 and M4 all function as switches, where a digital gate input decides whether the transistor should act as a short-circuit between drain and source, or as an open circuit. Since these transistors simply should have two possible states, it is key that the leakage current is minimized. This is particularly important for M1 and M2 to ensure that the voltage over  $C_s$  is as constant as possible during readout. In *Analog Circuit Design* by Tony Chan Carusone, one can read in section 1.4.1 that the subthreshold leakage current is given by

$$I_{off} = (n - 1)\mu_n C_{ox} \left(\frac{W}{L}\right) \left(\frac{kT}{q}\right)^2 \exp(-qV_t/nkT). \quad (2)$$

In order to minimize this leakage current,  $\frac{W}{L}$  need to be as small as possible, meaning the smallest possible width  $W$  and the largest possible length  $L$  is desired. The technology used will limit the possible width and length of the transistor, thus  $W$  and  $L$  can be chosen accordingly.

### 2.1.3 Value for $C_s$

To choose a suitable value for  $C_s$  spice simulations can be used. To know what values are the best we will look at the four corner cases for exposure time and light conditions. The corner cases will be denoted exposure-light, so for example max-min is maximum exposure time and minimum light. The voltage over  $C_s$  after exposure in each corner will get different values for different values of  $C_s$ . There are many possible approaches to how these corners should be tuned, and the one that we will apply is the following:

- Max-max corner should make  $C_s$  fully charged.
- Min-min corner should leave  $C_s$  uncharged.
- Min-max and max-min corners should make  $C_s$  half full charged.

The reason for why we want the corners to be like that is beyond the scope of this report.

#### 2.1.4 Values for M3 and active load




Figure 3: Simplified schematic for output voltage driving circuit.

Values for M3 and active load should be tuned for maximum dynamic range, but it will not have a very big impact since it is a very robust and self-regulating system. If we assume  $C_{C1}$  and  $C_{C2}$  are very small, the output system can be simplified to figure 3. Essentially, the output voltage  $V_{OUT}$  depends on how the conductance in the uppermost and downmost transistor are relative to each other. These conductances are proportional to  $W/L$  in the transistors and they also vary with the gate-source voltages. To avoid unnecessary simplifications, it is best to decide these  $W/L$ -ratios from SPICE-simulations. The dynamic range is the difference between the output voltage when uncharged and fully charged, and it is preferable that it is as big as possible.

#### 2.1.5 Process variations

It is important to look at how process variations will impact an analog design. One way to do this is to look at the what is called FF, FS, SF and SS corners. F means fast, S means slow, the first letter is for NMOS and the second is for PMOS. For this design it is enough to look at how the switch transistors are affected by process variations. The values to measure in each corner is  $R_{DS}$  when switched off and on.

For curiosity we will also have a look at how these two corners affect the analog circuit as a whole.



## 2.2 Digital camera controller

The digital input signals in the analog circuit need to be controlled by a digital circuit - the digital camera controller. The module interface for the digital camera controller is described in the figure below.




Figure 4: Overview of input and output pins for the camera controller module.

It has inputs *init*, *exp\_increase*, *exp\_decrease*, *clk* and *rst*. *clk* is controlled by an internal camera module, while the four other signals are controlled by the user. *init* is **HIGH** when the user presses the shutter button, *exp\_increase* is **HIGH** when the user wants to increase the exposure time, *exp\_decrease* is **HIGH** when the user wants to decrease the exposure time (and is overridden by *exp\_increase* if they conflict, i.e. if they are **HIGH** at the same time), and *rst* is **HIGH** when the user wishes to cancel any ongoing process, or reset the exposure time.

The outputs are *NRE\_1*, *NRE\_2*, *ADC*, *EXPOSE* and *ERASE*, all of which were described in section 2.1.

### 2.2.1 Finite state machine

The wanted behaviour of the digital camera controller is described in the finite state machine (FSM) in figure 5.




Figure 5: The finite state machine for the digital camera controller.

In the FSM, there are three states. The camera starts in the *idle* state. In this state, the camera isn't in the process of taking any picture, so output signals *EXPOSE* and *ADC* should be **LOW**, and *NRE\_1* and *NRE\_2* should be **HIGH**. The camera should also prepare itself for a new picture in this state, so *ERASE* should be **HIGH** to erase any voltage from the potential previous exposure. The user should be able to adjust the exposure time with the signals *exp\_increase* and *exp\_decrease* in the *idle* state. These signals will respectively increase or decrease the exposure time with 1ms for each clock cycle they're **HIGH** to a minimum of 2ms and a maximum of 30ms. When the signal *init* is **HIGH**, the camera will enter the *exposure* state. The exception is if the *rst* input signal is **HIGH**. In any state, *rst* being **HIGH** should override any process and take the camera back to *idle*.

When the camera is entering the *exposure* state, *ERASE* should be set **LOW** and *EXPOSE* should be set **HIGH** to begin the exposure. If *rst* is **HIGH** at any time during *exposure*, the camera should return to the *idle* state. When the preset exposure time has passed, the camera should enter the *readout* state if *rst* is **LOW**. Exposure being finished is called *exp\_done* in figure 5.

The *readout* state is the final state, in which the data from the individual pixels are read and inputted in the ADC. First, *EXPOSE* should be set **LOW** since the exposure is done. Then, *NRE\_1* should be **LOW** first at the same time as *ADC* is **HIGH**, to read from the top two pixels and input it to the ADC. Lastly, *NRE\_2* should be **LOW** at the same time as *ADC*

is **HIGH** to do the same for the bottom two pixels. Exactly how the output signals should be will be properly explored in the next section. When the readout sequence is finished, called *ro\_done* in figure 5, the camera should enter the *idle* state again. As with the *exposure* state, *rst* being **HIGH** at any time should return the camera to *idle*. It is important that the camera stays in *idle* for at least one clock cycle before the user is able to take another picture, to ensure that the data of the previous photo is erased.

## 2.2.2 Output timing

The desired time chart is illustrated in figure 6.




Figure 6: The desired time chart for the output signals of the digital camera controller.

In the *idle* state and *exposure* state, the output signals are as described in the previous subsection. Notably, *ERASE* is **HIGH** during *idle* and *EXPOSE* is **HIGH** during *exposure*. The *readout* state is more complex, as output signals are altered in every clock cycle. In the first cycle, T<sub>3</sub> in the figure, *EXPOSE* is set **LOW** to end exposure before readout begins. From T<sub>4</sub> to T<sub>6</sub>, *NRE\_1* is set **LOW**, with *ADC* being **HIGH** in the middle T<sub>5</sub>. This is to ensure that *NRE\_1* is not transienting from **HIGH** to **LOW** or **LOW** to **HIGH** while the ADC is reading. In T<sub>7</sub>, *NRE\_1* is set **HIGH** again. From T<sub>8</sub> to T<sub>10</sub>, the reading process repeats for *NRE\_2* to read the bottom two pixels. In the last clock cycle of the state, T<sub>11</sub>, *NRE\_2* is set **HIGH** again to end the readout. It can be noted that the *readout* state has a fixed duration of 9 clock cycles, as the only state with a fixed duration.

Table 1: Chosen component values based on simulations. N/A means not applicable.

Component	$W$	$L$	$C$
M1	$1.08\mu\text{m}$	$1.08\mu\text{m}$	N/A
M2	$1.08\mu\text{m}$	$1.08\mu\text{m}$	N/A
M3	$5.04\mu\text{m}$	$0.36\mu\text{m}$	N/A
M4	$1.08\mu\text{m}$	$1.08\mu\text{m}$	N/A
MC1	$1.08\mu\text{m}$	$1.08\mu\text{m}$	N/A
MC2	$1.08\mu\text{m}$	$1.08\mu\text{m}$	N/A
$C_S$	N/A	N/A	2pF

## 3 Results

### 3.1 Analog pixel circuit

#### 3.1.1 Values and dimensions

The transistor technology used limits the width to

$$1.08\mu\text{m} \leq W \leq 5.04\mu\text{m} \quad (3)$$

and limits the length to

$$0.36\mu\text{m} \leq L \leq 1.08\mu\text{m}. \quad (4)$$

As explained in section 2.1.2,  $W/L$  should be as small as possible to reduce leakage current. Therefore the length and width of the switch transistors will be  $1.08\mu\text{m}$ .

$C_S$  was tuned to 2pF. The output from simulations with this value for the exposure-light corners min-min, min-max, max-min and max-max are shown respectively in figures 7, 8, 9 and 10 in appendix A. These simulations were done by transient analysis of the voltage over  $C_S$  while applying photodiode current and exposure time for the corner cases. The resulting voltage over  $C_S$  is the value at the end of the exposure time.

For M3 and the active load the dynamic range was largest when  $W/L$  for M3 was as large as possible and  $W/L$  for MC1 and MC2 as small as possible. The simulation used for this is as follows: Simulate a transient analysis on the netlist for the full analog circuit, set *NRE\_1* LOW and *NRE\_2* HIGH. *ERASE* should first be HIGH a little while, then when it gets LOW, *EXPOSE* goes HIGH, and it should be HIGH long enough for the voltage on the OUT wires to reach its maximum value. The dynamic range is then the difference between the maximum and minimum of this curve. The graph from this simulation with final value is shown in figure 11 in appendix A.



it reaches 30 at the 15th cycle and stays constant. This shows that the exposure time will not exceed 30ms, as specified in section 2.2.1. It has also been made a point from clock cycle seven to nine that *exp\_increase* overrides *exp\_decrease* where they collide, a behaviour which also was specified as intended in section 2.2.1.

In the 17th clock cycle, *init* is HIGH, which sends the camera into the *exposure* state. The output signals act accordingly as *ERASE* goes LOW and *EXPOSE* goes HIGH. *cnt* is set to 29 and begins decrementing for each clock cycle. The *exposure* state is stopped short at cycle 22, where *rst* goes high for one cycle. As desired, this returns the camera to the *idle* state and sets *exp\_time* back to its default, 16.

Then, from cycle 24 to 39, *exp\_decrease* is set LOW. *exp\_time* decrements as intended, and stays constant after it reaches the minimum of 2. In the 40th cycle, *init* is HIGH again, causing the camera to enter the *exposure* state. *cnt* is set to 1, and decrements for one cycle, after which it has reached 0. True to the FSM illustrated in figure 5, this prompts the camera to enter the *readout* state.

The *cnt* register is set to 8 as the camera switches to *readout*. It rightfully decrements for each clock cycle, and the output signals act exactly as specified for the *readout* state in figure 6. When the readout is finished, i.e. after *cnt* has reached 0, the camera enters the *idle* state again.

## 4 Discussion

### 4.1 Analog

#### 4.1.1 Process variation

The values for  $R_{DS}$  for a switch transistor turned off is  $5\text{G}\Omega$  in FF,  $500\text{G}\Omega$  in TT and  $50\text{T}\Omega$ . For a voltage of 1V this means a leakage of 200pA in FF, and that is actually quite dramatic in our circuit. In SS however, a voltage of 1V means a leakage current of 20fA, and that is actually better for our circuit than the TT case.

The  $R_{DS}$  values when the transistors are switched on, however are not varying that much. The  $R_{DS}$  values for FF, TT and SS respectively are  $500\text{k}\Omega$ ,  $750\text{k}\Omega$  and  $1\text{M}\Omega$ . To conduct a current of 1nA, the required voltages are respectively  $500\mu\text{V}$ ,  $750\mu\text{V}$  and 1mV. All these values are very small compared to  $V_{DD}$ , so when switched on, our transistors can be considered ideal conductors for practical purposes.

For the full analog circuit, we can see as expected that the way in which the FF corner increases leakage current has a dramatic impact on the system as a whole, while the way SS increases the on resistance does not make a big change. And the cool thing is that SS also reduces leakage current compared to TT, so that SS in total improves behaviour of our design.

## 5 Summary

There are many considerations to take when designing a digital camera. In this report we have looked at two important systems, the camera controller and the pixel exposure circuit (in this report called “the analog circuit”). The analog circuit has been tested using SPICE simulations (using the program AIM-Spice) and the digital camera controller has been written in Verilog code and simulated in the program ActiveHDL. For the analog part we have also performed an analysis of what will happen to the switch transistors for some different process variations.

# Appendices

## A Analog simulation graphs




Figure 7: Minimum exposure time - minimum light




Figure 8: Minimum exposure time - maximum light






Figure 9: Maximum exposure time - minimum light




Figure 10: Maximum exposure time - maximum light




Figure 11: Simulation for maximizing dynamic range of M3 and active load.




Figure 12: FF corner simulation of  $R_{DS}$  of NMOS switch as function of  $V_{GS}$ .  $V_{DS} = 1.8V$ .




Figure 13: TT corner simulation of  $R_{DS}$  of NMOS switch as function of  $V_{GS}$ .  $V_{DS} = 1.8V$ .




Figure 14: SS corner simulation of  $R_{DS}$  of NMOS switch as function of  $V_{GS}$ .  $V_{DS} = 1.8V$ .




Figure 15: Simulation of analog circuit. The color(node nr)'s are: red(2): EXPOSE, light green(3): ERASE, blue(4): NRE\_1, magenta(5): NRE\_2, light blue(6): OUT1, brown(7): OUT2, dark green(8):  $V_{C11}$ , mustard(9):  $V_{C12}$ , dark blue(10):  $V_{C21}$ , teal(11):  $V_{C22}$ .




Figure 16: FF corner simulation of full analog circuit. Colors are the same as in figure 15.




Figure 17: SS corner simulation of full analog circuit. Colors are the same as in figure 15.

## B Spice code

Listing 1: SPICE code for full analog simulation.

```
[ aimspace ]
[ description ]
2050
* Pixel subcircuit

.include model_files/p18_cmos_models_ss.inc
.include model_files/PhotoDiode.cir

.param minW 1.08u
.param maxW 5.04u
.param maxL 1.08u
.param minL 0.36u

.param M1W minW
.param M1L maxL
.param M2W minW
.param M2L maxL
.param M3W maxW
.param M3L minL
.param M4W minW
.param M4L maxL

.param MCW minW
.param MCL maxL
.param CC_value 3p

.param CS_value 2p

.subckt pixel VDD VSS EXPOSE ERASE NRE OUT N2
.param I.D 150p
xPhotoDiode VDD 1 PhotoDiode Ipd.1 = I.D
* How to make NMOS: MX drain gate source bulk NMOS W= L=
M1 1 EXPOSE N2 VSS NMOS W=M1W L=M1L
M2 N2 ERASE VSS VSS NMOS W=M2W L=M2L

* How to make capacitor: CX node1 node2 value
CS N2 VSS CS_value

* How to make PMOS: MX source gate drain bulk PMOS W= L=
M3 3 N2 VSS VDD PMOS W=M3W L=M3L
M4 OUT NRE 3 VDD PMOS W=M4W L=M3L
.ends

.subckt load VDD VSS OUT
MC VDD OUT OUT VDD PMOS W=MCW L=MCL
```

```

CC OUT VSS CC_value
.ends

* How to make pixel:
* xName      VDD  VSS  EXPOSE  ERASE  NRE  OUT   N2    pixel  I_D=
  xPixel11    1    0      2      3     4    6    8    pixel  I_D=750p
  xPixel12    1    0      2      3     4    7    9    pixel  I_D=50p
  xPixel21    1    0      2      3     5    6   10    pixel  I_D=300p
  xPixel22    1    0      2      3     5    7   11    pixel  I_D=100p

xLoad1 1 0 6 load
xLoad2 1 0 7 load

VDD 1 0 dc 1.8v
* How to make pulse:
* V1 – initial voltage; V2 – peak voltage; TD – initial delay time;
* Tr – rise time; Tf – fall time; pwf – pulse-wise; and Period – period.
* Vname  N1  N2  pulse(  V1    V2    TD    Tr    Tf    PW    Period)
  vERASE  3    0  pulse(  0v    1.8V  0m    1u    1u    1m    15m)
  vEXPOSE 2    0  pulse(  0v    1.8v  1m    1u    1u    5m    15m)
  vNRE1   4    0  pulse(  1.8v   0v    7m    1u    1u    3m    15m)
  vNRE2   5    0  pulse(  1.8v   0v   11m    1u    1u    3m    15m)

* vNRE1 4 0 dc 1.8v
* vNRE2 5 0 dc 0v

* How to make transient analysis: .TRAN TSTEP TSTOP
.plot TRAN V(2) V(3) V(4) V(5) V(6) V(7) V(8) V(9) V(10) V(11)

[tran]
100n
15m
X
X
0
[ana]
4 0
[end]

```

Listing 2: SPICE code for CS value simulation.

```

[aimspice]
[description]
2030
* Simulation to tune CS

.include model_files/p18_cmos_models_tt.inc
.include model_files/PhotoDiode.cir

```

```

.param minW 1.08u
.param maxW 5.04u
.param maxL 1.08u
.param minL 0.36u

.param M1W minW
.param M1L maxL
.param M2W minW
.param M2L maxL
.param M3W maxW
.param M3L minL
.param M4W minW
.param M4L maxL

.param MCW minW
.param MCL maxL
.param CC_value 3p

.param CS_value 2p

.subckt pixel VDD VSS EXPOSE ERASE NRE OUT N2
.param I_D 150p
xPhotoDiode VDD 1 PhotoDiode Ipd_1 = I_D
* How to make NMOS: MX drain gate source bulk NMOS W= L=
M1 1 EXPOSE N2 VSS NMOS W=M1W L=M1L
M2 N2 ERASE VSS VSS NMOS W=M2W L=M2L

* How to make capacitor: CX node1 node2 value
CS N2 VSS CS_value

* How to make PMOS: MX source gate drain bulk PMOS W= L=
M3 3 N2 VSS VDD PMOS W=M3W L=M3L
M4 OUT NRE 3 VDD PMOS W=M4W L=M3L
.ends

.subckt load VDD VSS OUT
MC VDD OUT OUT VDD PMOS W=MCW L=MCL
CC OUT VSS CC_value
.ends

* How to make pixel:
* xName      VDD  VSS  EXPOSE  ERASE  NRE  OUT   N2    pixel  I_D=
  xPixel11    1    0      2        3      4     6     8    pixel  I_D=750p
  xPixel12    1    0      2        3      4     7     9    pixel  I_D=50p
  xPixel21    1    0      2        3      5     6    10    pixel  I_D=300p
  xPixel22    1    0      2        3      5     7    11    pixel  I_D=100p

```

```

xLoad1 1 0 6 load
xLoad2 1 0 7 load

VDD 1 0 dc 1.8v
* How to make pulse:
* V1 – initial voltage; V2 – peak voltage; TD – initial delay time;
* Tr – rise time; Tf – fall time; pwf – pulse-wise; and Period – period.
* Vname  N1  N2  pulse(  V1  V2  TD  Tr  Tf  PW  Period)
  vERASE  3   0  pulse(  0v  1.8V  0m  1u  1u  1m  40m)
  vEXPOSE  2   0  pulse(  0v  1.8v  1m  1u  1u  30m  40m)
* vNRE1   4   0  pulse( 1.8v   0v   7m  1u  1u   3m  15m)
* vNRE2   5   0  pulse( 1.8v   0v  11m  1u  1u   3m  15m)

  vNRE1    4    0  dc    1.8v
  vNRE2    5    0  dc    1.8v

* How to make transient analysis: .TRAN TSTEP TSTOP
.plot TRAN V(8)
[tran]
100n
40m
X
X
0
[ana]
4 0
[end]

```

Listing 3: SPICE code for M3 and active load values simulation.

```

[aimspice]
[description]
2050
* M3 and active load simulation

.include model_files/p18_cmos_models_tt.inc
.include model_files/PhotoDiode.cir

.param minW 1.08u
.param maxW 5.04u
.param maxL 1.08u
.param minL 0.36u

.param M1W minW
.param M1L maxL
.param M2W minW
.param M2L maxL
.param M3W maxW
.param M3L minL

```

```

.param MW minW
.param ML maxL

.param MCW minW
.param MCL maxL
.param CC_value 3p

.param CS_value 2p

.subckt pixel VDD VSS EXPOSE ERASE NRE OUT N2
.param I_D 150p
xPhotoDiode VDD 1 PhotoDiode Ipd_1 = I_D
* How to make NMOS: MX drain gate source bulk NMOS W= L=
M1 1 EXPOSE N2 VSS NMOS W=M1W L=M1L
M2 N2 ERASE VSS VSS NMOS W=M2W L=M2L

* How to make capacitor: CX node1 node2 value
CS N2 VSS CS_value

* How to make PMOS: MX source gate drain bulk PMOS W= L=
M3 3 N2 VSS VDD PMOS W=M3W L=M3L
M4 OUT NRE 3 VDD PMOS W=M4W L=M3L
.ends

.subckt load VDD VSS OUT
MC VDD OUT OUT VDD PMOS W=MCW L=MCL
CC OUT VSS CC_value
.ends

* How to make pixel:
* xName      VDD  VSS  EXPOSE  ERASE  NRE  OUT   N2    pixel  I_D=
  xPixel11    1    0      2       3     4     6     8    pixel  I_D=750p
  xPixel12    1    0      2       3     4     7     9    pixel  I_D=50p
  xPixel21    1    0      2       3     5     6    10    pixel  I_D=300p
  xPixel22    1    0      2       3     5     7    11    pixel  I_D=100p

xLoad1 1 0 6 load
xLoad2 1 0 7 load

VDD 1 0 dc 1.8v
* How to make pulse:
* V1 – initial voltage; V2 – peak voltage; TD – initial delay time;
* Tr – rise time; Tf – fall time; pwf – pulse-wise; and Period – period.
* Vname  N1  N2  pulse(  V1    V2    TD    Tr    Tf    PW  Period)
  vERASE  3    0  pulse(  0v    1.8V  0m    1u    1u    1m   15m)
  vEXPOSE 2    0  pulse(  0v    1.8v  1m    1u    1u   13m   15m)
* vNRE1   4    0  pulse(  1.8v   0v    7m    1u    1u    3m   15m)
* vNRE2   5    0  pulse(  1.8v   0v   11m    1u    1u    3m   15m)

```



```

vNRE1      4      0      dc      0v
vNRE2      5      0      dc      1.8v

* How to make transient analysis: .TRAN TSTEP TSTOP
.plot TRAN V(2) V(3) V(6)

[tran]
100n
15m
X
X
0
[ana]
4 0
[end]

```

Listing 4: SPICE code for process variations simulation.

```

[aimspice]
[description]
352
* Corner simulations for camera

.include model_files/p18_cmos_models_ss.inc

.param minW 1.08u
.param maxW 5.04u
.param maxL 1.08u
.param minL 0.36u

* How to make NMOS:
* MX drain gate source bulk NMOS W=      L=
M1      1      2      0      0 NMOS W=minW L=maxL

* VX + - DC(value)
V1 1 0 DC( 1.8v)
V2 2 0 DC( 0v)

.plot rds(M1)

[dc]
1
V2
0
1.8
0.001
[ana]
1 0

```

[end]

## C Verilog code

Listing 5: Verilog code for camera controller.

```
//-----  
//  
// Title      : camera_controller  
// Design     : camera_controller  
// Author     : sigurdht  
// Company    : NTNU  
//  
//-----  
//  
// File       : C:\Users\Sigurd\OneDrive - NTNU\elsys\semester5\ic\ic_project\verilog  
// Generated  : Tue Nov 3 15:17:41 2020  
// From       : interface description file  
// By         : Itf2Vhdl ver. 1.22  
//  
//-----  
//  
// Description :  
//  
//-----  
`timescale 1 ns / 1 ps  
  
//{{ Section below this comment is automatically maintained  
// and may be overwritten  
//{module {camera_controller}}  
module camera_controller ( init ,exp_increase ,exp_decrease ,  
    clk ,rst ,NRE_1 ,NRE_2 ,ADC ,expose ,erase );  
  
//}} End of automatically maintained section  
  
input wire init ;  
input wire exp_increase ;  
input wire exp_decrease ;  
input wire clk ;  
input wire rst ;  
output reg NRE_1 ;  
output reg NRE_2 ;  
output reg ADC ;  
output reg expose ;  
output reg erase ;  
  
reg [4:0] exp_time;  
reg [4:0] cnt;  
reg [1:0] state;  
  
// -- Enter your statements here -- //
```

```

initial begin
    NRE_1 = 1;
    NRE_2 = 1;
    ADC = 0;
    expose = 0;
    erase = 1;

    exp_time = 5'd16;
    cnt = 5'd0;
    state = 2'd0;
end

always @(posedge clk) begin
    if (rst) begin
        // Reset: Go to idle state
        NRE_1 = 1;
        NRE_2 = 1;
        ADC = 0;
        expose = 0;
        erase = 1;

        exp_time = 5'd16;
        cnt = 5'd0;
        state = 2'd0;
    end else begin
        case (state)
            2'd0: begin
                // State is idle
                if (init) begin
                    // Go to exposure state
                    erase = 0;
                    expose = 1;
                    cnt = exp_time - 1;
                    state = 2'd1;
                end else begin
                    if (exp_increase) begin
                        if (exp_time < 30) exp_time++;
                    end else if (exp_decrease) begin
                        if (exp_time > 2) exp_time--;
                    end
                end
            end
            2'd1: begin
                // State is exposure
                if (cnt > 0) begin
                    cnt--;
                end else begin

```

```

        // Go to readout state
        expose = 0;
        cnt = 5'd8;
        state = 2'd2;
    end
end
2'd2: begin
    // State is readout
    if (cnt > 0) begin
        case (cnt)
            5'd8: NRE_1 = 0;
            5'd7: ADC = 1;
            5'd6: ADC = 0;
            5'd5: NRE_1 = 1;
            5'd4: NRE_2 = 0;
            5'd3: ADC = 1;
            5'd2: ADC = 0;
            5'd1: NRE_2 = 1;
        endcase
        cnt--;
    end else begin
        // Go to idle state
        erase = 1;
        state = 2'd0;
    end
end
endcase
end
end
endmodule

```

Listing 6: Verilog code for camera controller testbench.

```

//-----
//
// Title      : testbench
// Design     : camera_controller
// Author     : sigurdht
// Company    : NTNU
//
//-----
//
// File       : C:\Users\Sigurd\OneDrive - NTNU\elsys\semester5\ic\ic-project\verilog
// Generated  : Tue Nov 3 15:07:12 2020
// From       : interface description file
// By         : Itf2Vhdl ver. 1.22
//
//-----

```

```

//
// Description :
//
//-----
`timescale 1 ms / 1 us

//{{ Section below this comment is automatically maintained
// and may be overwritten
//{module {testbench}}
module testbench ();

//}} End of automatically maintained section

// -- Enter your statements here -- //

reg init;
reg exp_increase;
reg exp_decrease;
reg clk;
reg rst;

wire NRE1;
wire NRE2;
wire ADC;
wire expose;
wire erase;

camera_controller camera_controller1(init, exp_increase, exp_decrease,
    clk, rst, NRE1, NRE2, ADC, expose, erase);

initial begin
    clk = 0;
    rst = 0;
    exp_increase = 0;
    exp_decrease = 0;
    init = 0;
    #1
    exp_increase = 1;
    #5
    exp_decrease = 1;
    #3
    exp_decrease = 0;
    #7
    exp_increase = 0;
    init = 1;
    #1
    init = 0;
    #4

```

```

    rst = 1;
    #1
    rst = 0;
    #1
    exp_decrease = 1;
    #16
    exp_decrease = 0;
    init = 1;
    #1
    init = 0;
    #16
    $finish;
end

always begin
    #0.5
    clk = ~clk;
end

endmodule

```